


# EXCEL 365, MAJ 2021

[POWERPAGE.SE/EXCELERERA](https://powerpage.se/excelerera)



ACCESS: PERSONAL

VERSION: UE31

**INNEHÅLL**

Basix.....	3
Läsanvisning.....	3
Markera .....	3
Formel.....	6
Tabell .....	9
Beräknad sammanställning .....	14
Fortsatta beräkningar med spillområden .....	19
Spill både åt höger och nedåt.....	20
Använda data från dynamiska tabeller .....	22
Tecken som styr hur områden används.....	24
Skärning .....	25
Skillnaden mellan två listor .....	31
Vad har försvunnit? .....	31
Vad har tillkommit? .....	34
Skillnaden för alla värden .....	35
Unika värden ur två listor .....	38
Talformat .....	42
Beyond .....	47
Matrisberäkningar .....	47
INDEX(matris;rad;[kolumn]) .....	48
Statiska matriser {5;6;4;7}{4\7\11}.....	48

## BASIX

### LÄSANVISNING

Dynamiska matriser finns fn endast i Excel 365. Emellertid är bl a genvägar i materialet allmängiltig även för tidigare versioner.

Tangentbordskombinationer skrivs med de tangenter som avses men med ett plustecken emellan. Exempelvis Ctrl+R avser håll ned tangenten Ctrl och tryck under tiden på R.

Följaktligen avser Ctrl++ att hålla ned tangenten Ctrl och under tiden trycka på plustangenten.

Skift+Mellanslag avser att hålla ned Shift/stor bokstav och trycka på mellanslagstangenten.


"=" är det vanligaste sättet att tala om för Excel att här kommer en formel. I detta dokument exkluderas normalt lika med-tecknet då de olika formlerna kan användas i varandra medan lika med-tecknet endast initierar beräkningen såsom dess första tecken.

### MARKERA

#### CELL

Klicka på en cell så markeras den med en förstärkt ram och adressen visas i namnrutan uppe till vänster.

Pekaren är ett stort plustecken när den markerar celler.



#### RAD & KOLUMN

Markera rader och kolumner genom att klicka på rubriken.


Det går också att ändra en markering från exempelvis cell B4 ovan till hela kolumn B genom att trycka Ctrl+Mellanslag.

Piltangenter för att flytta aktiv cell och Ctrl+Mellanslag för att markera kolumn/kolumner är alternativa sätt att markera som kan komma väl till pass beroende på hur tabellerna är utformade.

Ctrl+Mellanslag markerar kolumner utifrån aktuell markering.

Skift+Mellanslag markerar rader utifrån aktuell markering.


Båda tillsammans markerar alla celler.



Markering av ett kalkylblads kolumn görs på bladets rubrikrad och bokstaven får en ljus färgad bakgrund när den svarta pilen kommer att markera hela bladkolumnen.

Observera att pekaren byter till en svart pil som pekar åt det område som kommer att markeras. Det är pekarens pilspets som räknas och kolumnbokstavens färgtoning förstärker att det är kolumnen som markeras.

Om du ska markera en kolumn i en tabell så flytta ned pekaren så att pilspetsen är på tabellrubrikens överkant – men inte så långt ned att pekaren blir ett plustecken och endast markerar en rubrik, "Country", i den enskilda cellen B1.



Det går att markera celler och kolumner även genom att använda piltangenterna. Flytta till en cell och tryck Ctrl+Mellanslag så markeras kolumnen.

Markeringen blir olika beroende på vilken eller vilka celler du markerat. Om det markerade området hör till en tabell så blir det skillnad också om cellen är på en rad i tabellen eller på tabellens rubrikrad.

Eftersom markeringen ändras kan det också bli skillnad om man upprepar tangenttryckningen.


## CELLER OCH BLAD

Ctrl+Vänsterklick	Splittrad markering. Markering som Semikolon ; (union)
Skift+Vänsterklick	Utökad markering. Markering som Semikolon ; (till och med)
Skif+Piltangent	Utökad markering
Ctrl+End	Markera sista cell
Ctrl+Home	Markera första cell. Anpassas om låsta fönsterrutor/freeze panes
Ctrl+Skif+Home	Utöka markering till första cell
Ctrl+Mellanslag	Utöka markering till hela kolumnerna
Skif+Mellanslag	Utöka markering till hela raderna
Ctrl+Skift+Mellanslag	Markera hela kalkylbladet. Ungefär som Ctrl+AAA
Ctrl+. (Punkt)	Aktivera hörncell i markeringen

Ctrl+Piltangent	Markera datakant. Jämför att markera ord i text (nästa tomrum)
Ctrl+Skift+Piltangent	Utöka markering till datakant

## INMATNING

Ctrl+Enter	Matar in med samma innehåll/formel i hela det markerade området
Ctrl+'	Upprepa <i>exakt</i> innehållet från den cell som <i>syns</i> ovanför
Ctrl+D	Kopierar nedåt. Om endast en cell i höjded markerats så kopieras den som är ovanför, oavsett om den syns eller ej (Fill Down). Om flera raders höjd i markeringen kopieras den översta raden nedåt.
Ctrl+R	Som Ctrl+D men kopierar åt höger
Ctrl+Skift+,	Datum
Ctrl+Skift+.	Tid




Ctrl+Skift+. sedan Mellanslag sedan Ctrl+Skift+. resulterar i datum och tid som inte uppdateras.

## INFOGA/TA BORT

Ctrl++	Infoga nya celler enligt markering
Ctrl+-	Ta bort celler enligt markering

Om markeringen är en rad eller kolumn så utförs kommandot direkt.

Om markeringen är celler blir det oklart om kommandot avser höjded eller sidled och det kommer därför en följdfråga.



## DRA OCH SLÄPP

Ctrl	Kopiera
Skift	Skifta plats
Ctrl+Skift	Kopiera och skjut in

... med höger musknapp Meny som ger valmöjlighet

## FORMEL

### REFERENS TILL EN ENSKILD CELL

Genom att klicka på en cell, skriva = och sedan klicka på en annan cell så byggs nedan att E2 = B2.

Det blir lite långt från adressen i namnrutan till själva formeln men texten i formelfältet, längst upp, visar precis att det som visas i E2 är lika med<sup>1</sup> det som finns i cell B2.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		=B2				
3	2018-09-27	Austria	300						
4	2018-10-25	Czech repu	388						
5	2018-11-21	Germany	20000						
6	2018-12-04	Bulgaria	500						
7	2018-12-09	Austria	4800						

Nedan visar F2 formeltexten i E2. Detta för att förtydliga i dessa skärmbilder.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		Portugal	=B2			
3	2018-09-27	Austria	300						
4	2018-10-25	Czech repu	388						
5	2018-11-21	Germany	20000						
6	2018-12-04	Bulgaria	500						
7	2018-12-09	Austria	4800						

Cell F2 har på startmenyn givits cellformatet "Förklarande text".

#### Data och modell



Cell E2 kan kopieras eller autofyllas nedåt och kommer då att hämta Austria, Czech Republic, Germany osv.

<sup>1</sup>"=" talar om för Excel att här kommer en formel. I detta dokument exkluderas normalt lika med-tecknet då de olika formlerna kan användas i varandra medan lika med-tecknet endast initierar beräkningen såsom dess första tecken.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		Portugal	=B2			
3	2018-09-27	Austria	300		Austria	=B3			
4	2018-10-25	Czech Rep	388		Czech Rep	=B4			
5	2018-11-21	Germany	20000		Germany	=B5			
6	2018-12-04	Bulgaria	500		Bulgaria	=B6			
7	2018-12-09	Austria	4800		Austria	=B7			

Formeln i F2 ändrar sig "relativt" med kopieringen; från att hämta värdet i cell B2 till att referera till B3, B4, B5 osv.

#### REFERENS TILL ETT OMRÅDE

Genom att ange att en (1) cell refererar till flera så kommer Excel fram ungefär årsskiftet 2020, att låta svaret i den enda cellen spilla över till så många celler som behövs för att visa allt!

Funktionen kom med automatik till prenumeranter av Microsoft 365 (hette då Office 365) men levererades inte till alla samtidigt.

Den blåskimrande ramen visar hela resultatområdet.

E2									
	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		Portugal	=B2:B249			
3	2018-09-27	Austria	300		Austria				
4	2018-10-25	Czech Rep	388		Czech Republic				
5	2018-11-21	Germany	20000		Germany				
6	2018-12-04	Bulgaria	500		Bulgaria				
7	2018-12-09	Austria	4800		Austria				

Det är emellertid **endast** den första cellen som innehåller formeln. Resultatet har spillt till resterande celler och på formelfältet så är formeln samma som i E2 men den visas med grå text och är bara förklarande av vad som visas. Det går inte att redigera eller ändra formeln här. Det görs alltid "längst upp till vänster".

E5									
	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		Portugal	=B2:B249			
3	2018-09-27	Austria	300		Austria				
4	2018-10-25	Czech Rep	388		Czech Republic				
5	2018-11-21	Germany	20000		Germany				
6	2018-12-04	Bulgaria	500		Bulgaria				
7	2018-12-09	Austria	4800		Austria				


Skulle någon cell i resultatområdet vara i bruk så att svaret inte kan spilla över till så många celler som behövs, så blir det spillfel.

I bilden nedan finns det en text i spillområdet, i cell E5, och det blir därför spillfel i E2.

Det gröna hörnet och dess utropstecken meddelar att;

*"En cell som vi behöver spilla till är inte tom"*

\*

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000		#SPILL!	=B2:B249			
3	2018-09-27	Austria	300						
4	2018-10-25	Czech Rep	388						
5	2018-11-21	Germany	20000		Art. 32 - Insufficient technical and organisational meas				
6	2018-12-04	Bulgaria	500						
7	2018-12-09	Austria	4800						



## TABELL

Genom att klicka i ett område med data och på menyn Infoga: Tabell<sup>2</sup> aktiveras viss automatik för det aktuella dataområdet. Det är inte nödvändigt att markera hela området – Excel utökar markeringen till att omfatta hela dataområdet.



Tabellens sista cell, nedan C5, markeras med några extra färgpixlar som kan dras i för att ändra storleken på tabellen. Det är å andra sidan oftast inte nödvändigt eftersom tabellen själv expanderar när man skriver något på raden närmast nedanför, eller i kolumnen närmast till höger om.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000						
3	2018-09-27	Austria	300						
4	2018-10-25	Czech Repu	388						
5	2018-11-21	Germany	20000						
6									
7									

För att beräkna året av Datum, markera cell D3 och påbörja en formel genom att skriva =ÅR() och flytta in markören mellan parenteserna.

*(Det är en god vana att alltid sätta båda parenteserna samtidigt så minskar risken att det saknas någon. Vill du flytta in markören mha piltangenterna behöver du först trycka på F2. Piltangenterna har i princip 2 lägen – Skriv som innebär att man matar in och byter cell eller Redigera som innebär att markören flyttas inom formeln. Längst ned till vänster på statusraden visas vilket läge markören har.)*

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount						
2	2018-07-17	Portugal	400000						
3	2018-09-27	Austria	300	=ÅR([@Datum])					
4	2018-10-25	Czech Repu	388	ÅR(serienummer)					
5	2018-11-21	Germany	20000						
6									
7									

Klicka på A3 så fyller Excel i [@Datum]. Hakparenteserna markerar att det är ett fält och snabel-a att formeln inte avser hela fältet utan endast den cell som är på samma rad, dvs skärningspunkten.

Samma formel används i hela vägen, och kallas kolumnformel (liksom ovan visualiserat med en separat formel brevid som förklarar vilken formel som finns i cellen).

<sup>2</sup> För att terminologiskt skilja den här sortens tabell från ett vanligt område uppställt som en tabell, kan dessa kallas pyamas-tabeller utifrån att de som standard visas blårandiga.

Excel utökar inmatningen så den gäller för hela kolumn. Varje cell i tabellens kolumn får en kopia av formeln genom följande steg;

1. Formeln matas in i cell D3
2. Tabellen utökas med en pyramasmönstrad kolumn till
3. Formeln kopieras till alla celler i kolumnen

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Kolumr					
2	2018-07-17	Portugal	400000	2018	=ÅR([@Datum])				
3	2018-09-27	Austria	300	2018	=ÅR([@Datum])				
4	2018-10-25	Czech Repu	388	2018	=ÅR([@Datum])				
5	2018-11-21	Germany	20000	2018	=ÅR([@Datum])				
6									
7									

Om vi lägger till data på rad 6 så utökas tabellområdet och med samma automatik, även kolumnformeln.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	År					
2	2018-07-17	Portugal	400000	2018	=ÅR([@Datum])				
3	2018-09-27	Austria	300	2018	=ÅR([@Datum])				
4	2018-10-25	Czech Repu	388	2018	=ÅR([@Datum])				
5	2018-11-21	Germany	20000	2018	=ÅR([@Datum])				
6	2019-12-31	Germany	9550000	2019	=ÅR([@Datum])				
7									

Den beräknade kolumnen har ovan givits ett beskrivande namn samt cellformat som visar att den är beräknad.

### NB / Varning

Vanligen kan du göra en sak och sedan ångra den med Ctrl+Z. När det gäller kolumnformler så sker flera saker när du matar in dem och du kan se stegen var för sig genom att ångra – flera gånger.

Här måste du alltså ångra ett flertal gånger för att komma tillbaka till föregående läge!

När du lägger till en ny kolumn som ovan så är det väldigt tydligt vad som händer när du ångrar steg för steg.

Om du emellertid endast uppdaterar en befintlig kolumnformel så blir det bara två steg;

- a. formeln matas in och
- b. kopieras till alla celler i kolumnen med en kolumnformel

Steg 2 ovan, att utöka tabellen, utgår iom att kolumnen redan finns. Om du ångrar från b till a så har du en (1) cell som har ändringen men som inte kopierats ut och följaktligen inte stämmer överens med kolumnformeln. Ett litet grönt hörn varnar för att cellen har ett problem.

	Datum	Country	Amount	Share	SEK	F	G	H	I
241	2020-04-30	The Nether	725000	0,2%	7250000				
242	2020-05-05	Romania	5000	0,0%	50000				
243	2020-05-12	Sweden	11200	0,0%	123200				
244	2020-05-15	Denmark	6700	0,0%	67000				
245	2020-05-17	Ireland	75000	0,0%	750000				
246	2020-05-22	Finland	100000	0,0%	1000000				
247	2020-05-22	Finland	12500	0,0%	125000				

Exemplet ovan har en ny kolumn som beräknar SEK med kolumnformeln  $=[@Amount] * 10$   
 Cell E243 ändrades sedan till kursen 11 och hela kolumnen uppdaterades till  $=[@Amount] * 11$   
 Ctrl+Z ångrar endast ett (1) steg, från b till a och endast E243 har nu växelkursen 11. Resten av kolumn räknar med tio. Därav varningen/det gröna hörnet på E243;  
 "Den här cellen stämmer inte överens med kolumnformeln"

Varningstexten kommer fram genom att markera cellen och sätta muspekaren över utropstecknet till vänster om cellen. I bilden nedan är formeltexten återigen visad i en egen kolumn, som förklarande text.

	Datum	Country	Amount	Share	SEK	F	G	H	I
241	2020-04-30	The Nether	725000	0,2%	7250000	$=[@Amount]*10$			
242	2020-05-05	Romania	5000	0,0%	50000	$=[@Amount]*10$			
243	2020-05-12	Sweden	11200	0,0%	123200	$=[@Amount]*11$			
244	2020-05-15	Denmark	6700	0,0%	67000	$=[@Amount]*10$			
245	2020-05-17	Ireland	75000	0,0%	750000	$=[@Amount]*10$			
246	2020-05-22	Finland	100000	0,0%	1000000	$=[@Amount]*10$			
247	2020-05-22	Finland	12500	0,0%	125000	$=[@Amount]*10$			

Utropstecknet är också en meny där du kan få viss hjälp.



## KOLUMNER I TABELLER

En kolumn i kalkylbladet har alltid samma mängd rader och beror närmast på vilken version av Excel som används. I tidiga kalkylprogram var det under tusen rader och nu har Excel passerat en miljon.

Kolumner inom tabeller anpassas automatiskt till tabellens storlek och gör att vi inte behöver bearbeta fler värden än nödvändigt.

## FORMLER SOM ANVÄNDER OCH RETURNERAR OMRÅDEN

Genom att kombinera de två sätten att referera till en kolumn, skärning respektive hela området, kan du exempelvis räkna ut varje rads andel av totalen. Värdet på samma rad, skärningspunkten, delat med summan av hela kolumnen.

Nedan med talformatet procent och en decimal.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%	=[@Amount]/SUMMA([Amount])				
3	2018-09-27	Austria	300	0,0%	=[@Amount]/SUMMA([Amount])				
4	2018-10-25	Czech Repu	388	0,0%	=[@Amount]/SUMMA([Amount])				
5	2018-11-21	Germany	20000	0,2%	=[@Amount]/SUMMA([Amount])				
6	2019-12-31	Germany	9550000	95,8%	=[@Amount]/SUMMA([Amount])				
7									

Genom att referera till en kolumn i en tabell, och göra det från en cell som inte är i tabellen, behöver Excel lägga till tabellens namn för att hitta rätt kolumn. Samma kolumnnamn kan användas i flera tabeller men inom varje tabell måste kolumnerna ha unika namn.

Cell G2 nedan har formeln =Table1[Country] vilket betyder att refererar till en tabell med namnet Table1 och i den, en kolumn som heter Country.

Resultatet spiller till lika många celler som rader i tabellen.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%			Portugal	=Table1[Country]	
3	2018-09-27	Austria	300	0,0%			Austria		
4	2018-10-25	Czech Repu	388	0,0%			Czech Republic		
5	2018-11-21	Germany	20000	0,2%			Germany		
6	2019-12-31	Germany	9550000	95,8%			Germany		
7									

Lägger vi nu till den funktionen UNIK så kommer varje land bara att visas en gång.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%			Portugal		
3	2018-09-27	Austria	300	0,0%			Austria		
4	2018-10-25	Czech Repu	388	0,0%			Czech Republic		
5	2018-11-21	Germany	20000	0,2%			Germany		
6	2019-12-31	Germany	9550000	95,8%					
7									

Ytterligare en funktion på detta, SORTERA, så har vi de unika i alfabetisk ordning.

G2					=SORTERA(UNIK(Table1[Country]))				
	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%			Austria		
3	2018-09-27	Austria	300	0,0%			Czech Republic		
4	2018-10-25	Czech Repu	388	0,0%			Germany		
5	2018-11-21	Germany	20000	0,2%			Portugal		
6	2019-12-31	Germany	9550000	95,8%					
7									

Vi ser nu att vi har fyra olika värden, G2 till och med G5. Men det är bara G2 som innehåller en formel.

Med funktionen ANTALV kan Excel berätta hur många celler i ett område som inte är tomma. Emellertid ger ANTALV(G2) svaret 1 eftersom G2 endast innehåller ett enda värde.

H2					=ANTALV(G2)				
	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%			Austria	1	
3	2018-09-27	Austria	300	0,0%			Czech Republic		
4	2018-10-25	Czech Repu	388	0,0%			Germany		
5	2018-11-21	Germany	20000	0,2%			Portugal		
6	2019-12-31	Germany	9550000	95,8%					
7									

Celler som spiller kan göra det åt både höger och nedåt. För att referera till även de spillda celler som kommer efter G2 anges G2#. Tecknet visualiserar att G2 sträcker sig både åt höger och nedåt...

H2					=ANTALV(G2#)				
	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	4,0%			Austria	4	
3	2018-09-27	Austria	300	0,0%			Czech Republic		
4	2018-10-25	Czech Repu	388	0,0%			Germany		
5	2018-11-21	Germany	20000	0,2%			Portugal		
6	2019-12-31	Germany	9550000	95,8%					
7									

Mer data i tabellen uppdaterar beräkningarna automatiskt. Tabellens fler rader innehåller i det här fallet fler länder och de räknas upp en i taget och i sorterad ordning i G2#

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	Share					
2	2018-07-17	Portugal	400000	0,1%			Austria	26	
3	2018-09-27	Austria	300	0,0%			Belgium		
4	2018-10-25	Czech Repu	388	0,0%			Bulgaria		
5	2018-11-21	Germany	20000	0,0%			Croatia		
6	2018-12-04	Bulgaria	500	0,0%			Cyprus		
7	2018-12-09	Austria	4800	0,0%			Czech Republic		
8	2018-12-17	Germany	5000	0,0%			Denmark		
9	2018-12-18	Hungary	3200	0,0%			Finland		
10	2018-12-20	Austria	2200	0,0%			France		
11	2018-12-31	Austria	1800	0,0%					
12	2018-12-31	Germany	20000	0,0%					

### BERÄKNAD SAMMANSTÄLLNING

Den pyramasmönstrade tabellen skulle kunna laddas automatiskt från bokföringen, internet eller annan datakälla som csv- eller andra Excel-filer.

Den unika och sorterade uppräknigen av värden ska nu användas till att sammanställa information ur tabellen. Cell F1 = rubriken i cell B2. Här gäller det att markera endast rubrikcellen – markeras hela tabellkolumnen eller hela kolumn B så blir det spillfel i F1 eftersom det inte finns utrymme att spilla nedåt.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country			
2	2018-07-17	Portugal	400000			Austria			
3	2018-09-27	Austria	300			Belgium			
4	2018-10-25	Czech Repu	388			Bulgaria			
5	2018-11-21	Germany	20000			Croatia			
6	2018-12-04	Bulgaria	500			Cyprus			
7	2018-12-09	Austria	4800			Czech Republic			

SUMMA.OMF(SUMMAOMRÅDE;VILLKORSOMRÅDE1;VILLKOR1) / SUMIFS

“Adderar de celler som anges av en given uppsättning villkor.”

SUMMA.OMF summerar ett område med värden men jämför en lika stor kolumn med ett villkor och summerar endast när de överensstämmer.

SUMMA.OMF klarar av att använda **flera** villkor i samma funktion – därav summa.omF som i flera<sup>3</sup>. I det här fallet räcker det emellertid med ett enda villkor.

<sup>3</sup> =SUMMA.OMF(Table1[Amount]; Table1[Country];"=" & F2#; Table1[Amount]; ">=" & 20000)

BLAD				=SUMMA.OMF(Table1[Amount];Table1[Country];F2)			
				SUMMA.OMF(summaområde; villkorsområde1; villkor1; [villkorsområde2; villkor2]; ...)			
1	Datum	Country	Amount			Country	
2	2018-07-17	Portugal	400000			Austria	untry];F2)
3	2018-09-27	Austria	300			Belgium	
4	2018-10-25	Czech Repu	388			Bulgaria	
5	2018-11-21	Germany	20000			Croatia	
6	2018-12-04	Bulgaria	500			Cyprus	
7	2018-12-09	Austria	4800			Czech Republic	

Medan formeln skrivs in visas hjälptexter och färgmarkeringar för respektive referens.

**F2** innehåller endast Austria och resultatet blir summan för det enskilda landet.

G2				=SUMMA.OMF(Table1[Amount];Table1[Country];F2)			
1	Datum	Country	Amount			Country	
2	2018-07-17	Portugal	400000			Austria	18009100
3	2018-09-27	Austria	300			Belgium	
4	2018-10-25	Czech Repu	388			Bulgaria	
5	2018-11-21	Germany	20000			Croatia	
6	2018-12-04	Bulgaria	500			Cyprus	
7	2018-12-09	Austria	4800			Czech Republic	

Genom att ändra referensen till att inkludera hela matrisen spiller även den här formeln ut till ett lika stort område.

Genom att referera till **F2#** beräknar Excel en separat summering per land.

G2				=SUMMA.OMF(Table1[Amount];Table1[Country];F2#)			
1	Datum	Country	Amount			Country	
2	2018-07-17	Portugal	400000			Austria	18009100
3	2018-09-27	Austria	300			Belgium	85000
4	2018-10-25	Czech Repu	388			Bulgaria	3206130
5	2018-11-21	Germany	20000			Croatia	0
6	2018-12-04	Bulgaria	500			Cyprus	121000
7	2018-12-09	Austria	4800			Czech Rep	14658

I den blåa tabellen står Country innan Amount.

Cell F1 hämtar sin rubrik genom formeln =Table1[ [#Rubriker];[Country]]

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country			
2	2018-07-17	Portugal	400000			Austria	18009100		
3	2018-09-27	Austria	300			Belgium	85000		
4	2018-10-25	Czech Repu	388			Bulgaria	3206130		
5	2018-11-21	Germany	20000			Croatia	0		
6	2018-12-04	Bulgaria	500			Cyprus	121000		
7	2018-12-09	Austria	4800			Czech Rep	14658		

Referensen är tämligen "absolut" – jämför med att sätta dollartecken framför en referens såsom =\$B\$1.

Den hanteras också på ett absolut sätt om;

- F1 kopieras och klistras in i G1
- F1:F2 markeras och fylls åt höger, Ctrl+R
- F1:F2 markeras och formeln matas in med Ctrl+Enter (kopierar formeln till hela det markerade området)

Återstår dock autofyll; Markera F1 som i bilden ovan och ta tag i cellens nedre högra hörn med en svart korspekare.

Excel hämtar nu värde för värde från tabellens rubrikrad, och nästa på tur är Amount.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount		
2	2018-07-17	Portugal	400000			Austria	18009100		
3	2018-09-27	Austria	300			Belgium	85000		
4	2018-10-25	Czech Repu	388			Bulgaria	3206130		
5	2018-11-21	Germany	20000			Croatia	0		
6	2018-12-04	Bulgaria	500			Cyprus	121000		
7	2018-12-09	Austria	4800			Czech Rep	14658		

### XLETAUPP(LETAUPPVÄRDE; LETAUPPVEKTOR; RESULTATVEKTOR) / XLOOKUP

"Söker efter ett intervall eller en matris för en matchning och returnerar motsvarande objekt från ett andra intervall eller en andra matris. Som standard används exakt matchning."

XLETAUPP är en vidareutvecklad version av ett antal olika leta-funktioner. Kanske framförallt kombinationen index och passa/index match.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount	Första	
2	2018-07-17	Portugal	400000			Austria	18009100	43370	
3	2018-09-27	Austria	300			Belgium	85000	43613	
4	2018-10-25	Czech Repu	388			Bulgaria	3206130	43438	
5	2018-11-21	Germany	20000			Croatia	0	43903	
6	2018-12-04	Bulgaria	500			Cyprus	121000	43763	
7	2018-12-09	Austria	4800			Czech Rep	14658	43398	



43370 dagar från förra sekelskiftet blir 27 sep 2018. Jämför cell H2 med cell A3. Talformatering spiller inte över och H-kolumnen behöver därför formateras som datum.

Tryck Ctrl+Mellanslag för att markera hela kolumn H och sedan Ctrl+1 för att formatera talen som datum.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount	Första	
2	2018-07-17	Portugal	400000			Austria	18009100	2018-09-27	
3	2018-09-27	Austria	300			Belgium	85000	2019-05-28	
4	2018-10-25	Czech Repu	388			Bulgaria	3206130	2018-12-04	
5	2018-11-21	Germany	20000			Croatia	0	2020-03-13	
6	2018-12-04	Bulgaria	500			Cyprus	121000	2019-10-25	
7	2018-12-09	Austria	4800			Czech Rep	14658	2018-10-25	

För att ta fram även det senaste datumet kan H kopieras till I. Ändra först formeln så att villkorsområdet i F2# blir absolut och inte följer med i kopieringen.

H2    =XLETAUPP(F2#;Table1[Country];Table1[Datum])

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount	Första	
2	2018-07-17	Portugal	400000			Austria	18009100	2018-09-27	
3	2018-09-27	Austria	300			Belgium	85000	2019-05-28	
4	2018-10-25	Czech Repu	388			Bulgaria	3206130	2018-12-04	
5	2018-11-21	Germany	20000			Croatia	0	2020-03-13	
6	2018-12-04	Bulgaria	500			Cyprus	121000	2019-10-25	
7	2018-12-09	Austria	4800			Czech Rep	14658	2018-10-25	

Dollartecken framför F gör att referensen fortsatt är F2 när H2 kopieras till i2.

Markera H1:i2 och tryck Ctrl+R som är kortkommandot för Start: Fyll: Höger/Fill Right.

XLETAUPP(lookup\_value; lookup\_array; return\_array; [if\_not\_found]; [match\_mode]; [search\_mode])

BLAD    =XLETAUPP(\$F2#;Table1[Country];Table1[Datum])

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount	Första	
2	2018-07-17	Portugal	400000			Austria	18009100	\$F2#;	
3	2018-09-27	Austria	300			Belgium	85000	2019-05-28	
4	2018-10-25	Czech Repu	388			Bulgaria	3206130	2018-12-04	
5	2018-11-21	Germany	20000			Croatia	0	2020-03-13	
6	2018-12-04	Bulgaria	500			Cyprus	121000	2019-10-25	
7	2018-12-09	Austria	4800			Czech Rep	14658	2018-10-25	

Återstår att ändra rubrik/etikett och lägga till en av de frivilliga parametrarna i XLETAUPP så att sökningen sker från sista till första.

XLETAUPP(letauppvärde; letauppvektor; resultatvektor; [om ej matchning]; [matchningsätt]; [söksätt])

- 1 - Sök första till sista
- 1 - Sök sista till första
- 2 - Binär sökning (sorteras i stigande ordning)

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount			Country	Amount	Första	Sista
2	2018-07-17	Portugal	400000			Austria	18009100	2018-09-27	2019-10-23
3	2018-09-27	Austria	300			Belgium	85000	2019-05-28	2020-05-29
4	2018-10-25	Czech Repu	388			Bulgaria	3206130	2018-12-04	2020-02-20
5	2018-11-21	Germany	20000			Croatia	0	2020-03-13	2020-03-13
6	2018-12-04	Bulgaria	500			Cyprus	121000	2019-10-25	2020-01-13
7	2018-12-09	Austria	4800			Czech Rep	14658	2018-10-25	2019-05-06

De parametrar som inte behöver ändras lämnas tomma och endast i den sista anges -1.

=XLETAUPP(\$F2#;Table1[Country];Table1[Datum];;-1)

Observera att söksätt = -1 innebär att det är sista träffen som returneras. Till skillnad mot tidigare leta-funktioner så behöver inte data vara sorterad!

### STABILISERA FORMELN

Vill vi säkert få fram högsta värdet så vore kombination av MAX och FILTER lockande, men MIN och MAX är inte spillande och kan därför inte användas till detta. Med Excel-version innan dynamiska matriser hade tabellen först behövt sorteras kronologiskt så att första och sista träff motsvarar min och max.

SORTERA(Table1;1) sorterar hela tabellen efter första kolumnen, som i det här fallet är datum.

XLETAUPP behöver emellertid söka i en (1) kolumn och returnera en (1) annan. Det kan göras med funktionen INDEX(matris;rad;[kolumn]) som normalt returnerar en enskild cell utifrån angiven rad och kolumn – tänk radnummer och kolumnnummer inom det angivna området/matrisen.

Genom att ange noll som radnummer så refererar funktionen inte till en enskild rad i en enskild kolumn utan returnerar istället alla rader i kolumnen. Kolumnen Datum kan på det viset refereras som INDEX(Table1;0;1) och kolumnen Country som INDEX(Table1;0;2).

För att få tabellen sorterad efter första kolumnen ersätts Table1 med SORTERA(Table1;1) och inlagt i XLETAUPP-formeln ovan, XLETAUPP(\$F2#;Table1[Country];Table1[Datum];;-1) blir den sorterad;

=XLETAUPP(\$M2#;INDEX(SORTERA(Table1;1);0;2);INDEX(SORTERA(Table1;1);0;1);;-1)

Bör noteras att formeln här är hårdkodad till att alltid ta kolumn ett som datum och två som land. Om man kan lita på att kolumnerna alltid kommer i samma ordning så är det en möjlig lösning. Om emellertid kolumnordningen kan komma att ändras så behöver den sista ettan, den som pekar ut datumkolumnen som den tabellens första, ersättas med att söka efter den kolumn som har rubriken Datum.

XMATCHNING("Datum";Table1[#Rubriker])

Formeln blir då ännu längre men oberoende av var i ordningen kolumnen Datum och Country finns;

=XLETAUPP(\$M2#;INDEX(SORTERA(Table1;1);0;XMATCHNING("Datum";Table1[#Rubriker]));INDEX(SORTERA(Table1;1);0; XMATCHNING("Country";Table1[#Rubriker]));;-1)

De hårdkodningar som nu återfinns är kolumnrubrikerna "Datum" och "Country". De i sin tur kan avslutningsvis anges via varsin cell på ett särskilt blad för inställningar, eg med namnet \_prefs. Delformeln skulle då kunna se ut som

XMATCHNING(\_prefs! !B75;Table1[#Rubriker])

## FORTSATTA BERÄKNINGAR MED SPILLOMRÅDEN

Cellerna H2 och I2 inkluderar båda två sina respektive spillområden om de refereras med # efter. Resultatet får samma behov av att spilla till andra celler och formeln för att se antalet dagar mellan sista och första datumet blir kort och gott  $I2 - H2$ .

	F	G	H	I	J	K	L	M	N
1	Country	Amount	Första	Sista	Dagar				
2	Austria	18009100	2018-09-27	2019-10-23	=I2#-H2#				
3	Belgium	85000	2019-05-28	2020-05-29					
4	Bulgaria	3206130	2018-12-04	2020-02-20					
5	Croatia	0	2020-03-13	2020-03-13					
6	Cyprus	121000	2019-10-25	2020-01-13					
7	Czech Rep	14658	2018-10-25	2019-05-06					

J2          =I2#-H2#

	F	G	H	I	J	K	L	M	N
1	Country	Amount	Första	Sista	Dagar				
2	Austria	18009100	2018-09-27	2019-10-23	391				
3	Belgium	85000	2019-05-28	2020-05-29	367				
4	Bulgaria	3206130	2018-12-04	2020-02-20	443				
5	Croatia	0	2020-03-13	2020-03-13	0				
6	Cyprus	121000	2019-10-25	2020-01-13	80				
7	Czech Rep	14658	2018-10-25	2019-05-06	193				

Är det endast antalet dagar som önskas så kan kolumn H och I döljas och man har en modell som visar hur den har byggts upp.

Vill man komprimera modellen så vet vi att formlerna i de två cellerna spilla, och genom att klicka på dem kan vi se vad respektive är lika med;

H2          =XLETAUPP(\$F2#;Table1[Country];Table1[Datum])

I2          =XLETAUPP(\$F2#;Table1[Country];Table1[Datum];;-1)

Antalet dagar får vi genom att  $J2 = I2 - H2$

Genom att byta ut  $I2$  mot sin formel och samma för  $H2$  får vi en enda men lite längre formel:

= XLETAUPP(\$F2#;Table1[Country];Table1[Datum];;-1)-XLETAUPP(\$F2#;Table1[Country];Table1[Datum])

H2    = XLETAUPP(\$F2#;Table1[Country];Table1[Datum];;-1) - XLETAUPP(\$F2#;Table1[Country];Table1[Datum])

	F	G	H	I	J	K	L	M	N	O
1	Country	Amount	Dagar							
2	Austria	18009100	391							
3	Belgium	85000	367							
4	Bulgaria	3206130	443							
5	Croatia	0	0							
6	Cyprus	121000	80							
7	Czech Rep	14658	193							

### SPILL BÅDE ÅT HÖGER OCH NEDÅT

För att summera valfria celler såsom B2 till och med B47 används SUMMA(B2:B47)

Kolon betyder "till och med" och behandlar alla 46 cellerna som ett enda område.

På samma sätt använder vi här formeln SORTERA(F2#:H2#) vilket, som angivet med #, inkluderar spillområdena, dvs hela matriserna.

För att sortera på exempelvis beloppen och i fallande ordning anges en tvåa för att ange att sorteringen ska göras efter andra kolumnen i dataområdet (matrisen), och -1 för fallande.

Nu kommer den här enda formeln att spilla åt båda höger och nedåt.

J2    =SORTERA(F2#:H2#;2;-1)

	F	G	H	I	J	K	L	M	N
1	Country	Amount	Dagar						
2	Austria	18009100	391		United Kingdom	315310200	162		
3	Belgium	85000	367		France	51100000	304		
4	Bulgaria	3206130	443		Italy	39418000	324		
5	Croatia	0	0		Germany	34737807	405		
6	Cyprus	121000	80		Austria	18009100	391		
7	Czech Rep	14658	193		Sweden	7083530	266		

## SKÄRNINGSPUNKTEN @

I tidigare versioner av Excel blev det automatiskt skärningspunkten om ett område angavs som referens. Exempelvis kunde man i cell H3 ovan kunnat använda formeln =B:B som innebär att man refererar till hela kolumn B.

I detta exempel skulle vi fått resultatet Austria som finns i skärningspunkten, på samma rad. Excel föreslår och lägger själv in @ i sådana formler om de byggs med tidigare versioner.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	År					
2	2018-07-17	Portugal	400000	2018	=ÅR([@Datum])				
3	2018-09-27	Austria	300	2018	=ÅR([@Datum])			Austria	=@B:B
4	2018-10-25	Czech Repu	388	2018	=ÅR([@Datum])				
5	2018-11-21	Germany	20000	2018	=ÅR([@Datum])				
6	2019-12-31	Germany	9550000	2019	=ÅR([@Datum])				
7									

Utan snabel-a kommer referensen att gälla hela kolumn B och Excel låter numera resultatet spilla ut i flera celler.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	År					
2	2018-07-17	Portugal	400000	2018	=ÅR([@Datum])				
3	2018-09-27	Austria	300	2018	=ÅR([@Datum])			#SPILL!	=B:B
4	2018-10-25	Czech Repu	388	2018	=ÅR([@Datum])				
5	2018-11-21	Germany	20000	2018	=ÅR([@Datum])				
6	2019-12-31	Germany	9550000	2019	=ÅR([@Datum])				
7									

Eftersom det saknas två celler längst ned – vi har ju två ovanför – så blir det spillfel. ”Vi kan inte spilla utanför kalkylbladets kanter.”

Om vi däremot flyttar formeln längst upp så hämtas hela kolumnen med en enda formel.

	A	B	C	D	E	F	G	H	I
1	Datum	Country	Amount	År				Country	=B:B
2	2018-07-17	Portugal	400000	2018	=ÅR([@Datum])			Portugal	
3	2018-09-27	Austria	300	2018	=ÅR([@Datum])			Austria	
4	2018-10-25	Czech Repu	388	2018	=ÅR([@Datum])			Czech Republic	
5	2018-11-21	Germany	20000	2018	=ÅR([@Datum])			Germany	
6	2019-12-31	Germany	9550000	2019	=ÅR([@Datum])			Germany	
7								0	
8								0	
9								0	
10								0	
11									
12									

## ANVÄNDA DATA FRÅN DYNAMISKA TABELLER

Microsoft kallar dem "Tabeller" men utifrån att det kan missförstås som ett vanligt område uppställt som en tabell, och att de som standard blir blå-vit-randiga så kan de även kallas för pyamas-tabeller.

De har fördelen att de automatiskt ändrar storleken alltefter omfattningen på indata.


De är flexibla i både sid- och höjddled.

Om du hämtar data från internet eller annan källa som du inte har kontroll över så är det tänkbart att antal rader, antal kolumner och även kolumnnamn kommer att ändras över tid.

Orange tabell, Table1, kommer från textfil med värdekolumnerna Sweden, UK och Spain.

Total summa för hela tabellen, i det här fallet alla tre länderna, beräknas i A2 med formeln **SUMMA(Table1)**. Summan i B2 är **SUMMA(Table1[Sweden])**. Den har sedan autofyllts åt höger genom att dra i kopieringshandtaget i cellens nedre högre hörn.

Cell D2 har därmed fått formeln **SUMMA(Table1[Spain])**. Formlerna pekar tydligt och i klartext på respektive lands kolumn. Om en sådan cell kopieras och klistras in på ett annat ställe kommer den fortfarande att summera samma land som innan. Den betar sig alltså som en absolut referens<sup>4</sup>.



	A	B	C	D	E	F
1						
2	Fine €	£ 324 844 300	7 064 830	315 310 200	2 469 270	
3						
4	Type	Sweden	UK	Spain		
5	Insufficient technical and organisational measures		315 310 200	379 500		
6	Insufficient legal basis for data processing	64 830		1 444 570		
7	Non-compliance with general data processing principles			358 400		
8	Insufficient fulfilment of data subjects rights	7 000 000		27 000		
9	Insufficient fulfilment of information obligations			259 800		
10	Lack of appointment of data protection officer					

Om tabellen utökas med en kolumn<sup>5</sup> så förskjuts referenserna vilket kan vara ett problem.

<sup>4</sup> Celladresser med \$ framför ändras inte vid kopiering. Exempelvis \$H\$2 istället för relativa H2. Gäller som standard även för namngivna referenser.

<sup>5</sup> Markera en cell i kolumnen Spain, tryck *Ctrl Mellanslag* en gång för att markera kolumnen i tabellen och sedan *Ctrl +* för att infoga en ny kolumn. Fyll i data i den nya kolumnen (Germany).



Spain	
379 500	
1 444 570	
358 400	
27 000	
259 800	

-> Ctrl Mellanslag -> Ctrl + ->

Kolumn1	Spain
	379 500
	1 444 570
	358 400
	27 000
	259 800

->

Germany	Spain
9 935 000	379 500
55 018	1 444 570
14 796 000	358 400
245 407	27 000
	259 800
61 000	

Nedan; Cell D2 summerar fortfarande Spain. Vore det en summering som ska visa summan någon annanstans och inte i tabellens närhet så är det förmodligen rätt att summan fortfarande visar Spain.

Det är också logiskt eftersom formeln ju uttryckligen summerar just [Spain] och inte de tabellvärden som finns nedanför i tabellen. Det är som en absolut referens.

	A	B	C	D	E	F
1		=SUMMA(Table1)	=SUMMA(Table1[Sweden])	=SUMMA(Table1[UK])	=SUMMA(Table1[Spain])	
2	<b>Fine €</b>	<b>£ 349 936 725</b>	<b>7 064 830</b>	<b>315 310 200</b>	<b>2 469 270</b>	
3						
4	<b>Type</b>	<b>Sweden</b>	<b>UK</b>	<b>Germany</b>	<b>Spain</b>	
5	Insufficient technical and organisational measures		315 310 200	9 935 000	379 500	
6	Insufficient legal basis for data processing	64 830		55 018	1 444 570	
7	Non-compliance with general data processing principles			14 796 000	358 400	
8	Insufficient fulfilment of data subjects rights	7 000 000		245 407	27 000	
9	Insufficient fulfilment of information obligations				259 800	
10	Lack of appointment of data protection officer			61 000		

Genom att ännu en gång autofylla från B2 ändra formeln gentemot de nu aktuella kolumnrubriken och formeln beräknar summan av tabellkolumnen rakt nedanför.

B2    =SUMMA(Table1[Sweden])

	A	B	C	D	E	F
1		=SUMMA(Table1)	=SUMMA(Table1[Sweden])	=SUMMA(Table1[UK])	=SUMMA(Table1[Germany])	=SUMMA(Table1[Spain])
2	<b>Fine €</b>	<b>£ 349 936 725</b>	<b>7 064 830</b>	<b>315 310 200</b>	<b>25 092 425</b>	<b>2 469 270</b>
3						
4	<b>Type</b>	<b>Sweden</b>	<b>UK</b>	<b>Germany</b>	<b>Spain</b>	
5	Insufficient technical and organisational measures		315 310 200	9 935 000	379 500	
6	Insufficient legal basis for data processing	64 830		55 018	1 444 570	
7	Non-compliance with general data processing principles			14 796 000	358 400	
8	Insufficient fulfilment of data subjects rights	7 000 000		245 407	27 000	
9	Insufficient fulfilment of information obligations				259 800	
10	Lack of appointment of data protection officer			61 000		

## TECKEN SOM STYR HUR OMRÅDEN ANVÄNDS

Excel använder olika tecken för att bestämma hur och eventuellt vilka delar av områden ska hanteras. Kolon, semikolon, snabel-a, mellanslag och hashtag. Även dollartecken i de fall en beräkning ska kopieras till flera celler.

### SUMMA(TAL1; [TAL2]; ...) / SUM

”Adderar samtliga tal i ett cellområde”

#### KOLON :

**Kolon** används för ”till och med” så SUMMA(D5:D9) summerar cellerna D5 till och med D9, dvs fem celler. I tabellen blir summan 2 469 270.

Kolon kan jämföras med att med musen hålla ned skift (stor bokstav) och klicka på en cell. Markeringen blir från och med den första cellen till och med den sista.

Skift-tangenten utökar markeringen och gör så även i kombination med piltangenterna.

	A	B	C	D
1				=SUMMA(D5:D9)
2				2 469 270
3				
4	Type	Sweden	UK	Spain
5	Insufficient technical		315 310 200	379 500
6	Insufficient legal basis	64 830		1 444 570
7	Non-compliance with			358 400
8	Insufficient fulfilment	7 000 000		27 000
9	Insufficient fulfilment			259 800

#### SEMIKOLON ;

**Semikolon** används till att ange splittrade områden (union). Jämför att med musen hålla ned Ctrl-tangenten (control) och markera ytterligare ett område.

SUMMA(D5;D9) blir endast två celler, D5 och D9 och summan blir 639 300.

Formeltipset som Excel visar anger SUMMA(tal1; [tal2]; [tal3]; ...)

Betydelsen är att även om ”tal1” är det som närmast uteslutande används, så *kan* ytterligare tal eller områden anges i samma formel.

#### KOLON OCH SEMIKOLON

**Kolon och semikolon** kan också kombineras.

Summan för både Sweden och Spain kan skrivas som SUMMA(B5:B9; D5:D9) och blir 9 534 100.

Resultatet är detsamma som SUMMA(B5:B9) + SUMMA(D5:D9) men med en kortare formel.

	A	B	C	D
1				=SUMMA(B5:B9; D5:D9)
2				9 534 100
3				
4	Type	Sweden	UK	Spain
5	Insufficient technical		315 310 200	379 500
6	Insufficient legal basis	64 830		1 444 570
7	Non-compliance with			358 400
8	Insufficient fulfilment	7 000 000		27 000
9	Insufficient fulfilment			259 800



NB:

Semikolon kan i det här fallet ses som flertydigt i och med att det både avgränsar mellan parameter ett och två, samtidigt som det betyder ett splittrat område som består av område 1 och område 2 (tal1 och tal2). När det gäller funktioner som summa är semikolon som skiljetecken<sup>6</sup> mellan parametrar logiskt och liktydigt med union, dvs det har ingen skild betydelse.

## SKÄRNING

### SNABEL-A @

I tabeller används med automatik **snabel-a** för att markera skärningspunkten för en kolumn, och formeln använder då det värde i kolumnen som finns på samma rad som formeln (intersection).

	A	B	C	D	E	F
1	Date	Country	Sum Fine €	Year		
2	17 jul 2018	Portugal	400000	2018	=ÅR({@Date})	
3	27 sep 2018	Austria	300	2018	=ÅR({@Date})	
4	25 okt 2018	Czech Republic	388	2018	=ÅR({@Date})	
5	21 nov 2018	Germany	20000	=ÅR({@Date})		
6	31 dec 2019	Germany	497800	2019	=ÅR({@Date})	

Snabel-a fungerar och behövs även för referenser till exempelvis hela rader eller kolumner.

### MELLANSLAG

**Mellanslag** som skiljetecken har funnits sedan lång tid tillbaka och betyder **skärning**, dvs där två olika områden överlappar varandra (intersection).

SUMMA(B5:D5 D5:D9) summerar endast en cell, nämligen D5 som är gemensam för båda områdena. Skärningen uppstår genom att det är ett mellanslag mellan B5:D5 och D5:D9.

	A	B	C	D
1				=SUMMA(B5:D5 D5:D9)
2				379 500
3				
4	Type	Sweden	UK	Spain
5	Insufficient technical		315 310 200	379 500
6	Insufficient legal basis	64 830		1 444 570
7	Non-compliance with			358 400
8	Insufficient fulfilment	7 000 000		27 000
9	Insufficient fulfilment			259 800

Skärning används nedan till att med relativ funktionalitet summera kolumnerna i en tabell. I och med att de normalt refereras till med sina namn hanteras de annars som absoluta referenser – utom vid autofyll.

### NOTERVÄRT OM SUMMERING AV SKÄRNINGAR

I och med att Excel numera hanterar områden på ett mycket bättre sätt blir det viktigare att förstå hur dynamiska områden kan komma att påverka varandra.

<sup>6</sup> Semikolon används vanligen som skiljetecken i svenska csv-filer. Detta trots att csv står för comma separated values och anledningen är att med svensk standard separeras uppräkningsrader med semikolon – inte med kommatecken som ju redan används som decimaltecken.

Denna avvikelse ställer till problem för användare som nyttjar csv-filer från andra system och sedan, efter en smärre uppdatering, plötsligt inte längre får data i skilda kolumner. Hanteringen av detta problem beskrivs närmare i **Kommaseparerade filer, csv**.

De två ovanstående områdena, blått och rött, skär varandra i cell D5.  
 SUMMA(B5:D5 D5:D9) med mellanslag emellan resulterar i endast 379 500.

SUMMA(B5:D5) + SUMMA(D5:D9) resulterar i att D5 finns med i båda summorna och dubbelräknas.  
 315 689 700 + 2 469 270 = 318 158 970

SUMMA(B5:D5; D5:D9) har D5 finns med i båda talen och dubbelräknas även här. 318 158 970

**Statusraden exkluderar** emellertid dubbelräkningar! Markering av de två områdena visar summan 317 779 470.

Type	Sweden	UK	Spain	
Insufficient technical		315 310 200	379 500	=SUMMA(B5:D5)+SUMMA(D5:D9) 318 158 970
Insufficient legal basis	64 830		1 444 570	
Non-compliance with			358 400	=SUMMA(B5:D5; D5:D9) 318 158 970
Insufficient fulfilment	7 000 000		27 000	
Insufficient fulfilment			259 800	

Vill man räkna bort skärningar som dubbelräknata kan man subtrahera dem från summan.

SUMMA(B5:D5; D5:D9)-SUMMA(B5:D5 D5:D9) neutraliserar dem.

SUMMA(B5:D5; D5:D9; -B5:D5 D5:D9) om man vill komprimera formeln genom att använda tal3.

## HASHTAG #

En formel som spiller kan spilla flera kolumner och rader, åt höger och nedåt.

Cell E8 innehåller en formel som refererar till ett helt område, A1:C5, och resultatet spiller därför till ett lika stort område – fem rader och tre kolumner.

En referens till E8 adresserar endast den cellen (1) medan E8# inkluderar även de celler som kommer efter – de som formeln spillt till.

	A	B	C	D	E	F	G	H
1		pcs	Value					
2	Alfa	47	517					
3	Bravo	11	121					
4	Charlie	42	462					
5	Delta	19	209					
6								
7								
8					E8=A1:C5	pcs	Value	
9					Alfa	47	517	
10					Bravo	11	121	
11					Charlie	42	462	
12					Delta	19	209	
13								

## DOLLARTECKEN \$

\$ framför en referens låser den så den inte följer med vid kopiering. Ungefär som dollarn i Excels barndom var en fast växelkurs.

Formeln = $\$A\$1$  kan kopieras till annan cell men kommer fortsatt att referera till innehållet i A1.

Formeln =A1 kommer att följa med kopieringen och hela tiden hämta relativt till cellen med formeln.

Vid inmatning av formeln kan F4 användas för att växla mellan de fyra olika varianterna;

A1,  $\$A\$1$ ,  $\$A1$  och  $A\$1$ .

---

## RELATIV REFERENS AV TABELLKOLUMN

Den blå tabellen nedan heter Table1, och den gröna Table2.

Tabellnamn fungerar som andra definierade namn men med skillnaden att området är spärrat så det går inte att ändra i Namnhanteraren<sup>7</sup>. Detta eftersom tabeller själva ändrar sina områden/storlekar.

Användbarheten är emellertid densamma som med manuellt definierade eller skapade namn.

Summering av hela tabellen kan göras med SUMMA(Table1) och den blåa tabellen nedan helsummeras på det sättet i cell A2.

Table1[Sweden], dvs kolumnen Sweden i tabellen Table1, skulle kunna summeras med SUMMA(B5:B9). Eftersom den formeln inte har några dollartecken är den relativ och kan kopieras till kolumn C och D och då summera C5:C9 respektive D5:D9.

SUMMA(Table1[Sweden]) fungerar också i första fallet men om den bara kopieras och klistras in eller fylls åt höger så kommer alla kopior att summera Sweden.

Undantaget är om cellen autofylls åt höger. Då byts kolumnen namn till nästa i tabellen.

---

<sup>7</sup> Meny Formler: Namnhanteraren / Ctrl F3

Skärning kan användas till att med relativ funktionalitet summera kolumnerna i en tabell.

	A	B	C	D	E
1	=SUMMA(Table1)	=SUMMA(Table1 B:B)	=SUMMA(Table1 C:C)	=SUMMA(Table1 D:D)	
2	<b>Fine € Σ 324 844 300</b>	<b>7 064 830</b>	<b>315 310 200</b>	<b>2 469 270</b>	
3					
4	<b>Type</b>	<b>Sweden</b>	<b>UK</b>	<b>Spain</b>	
5	Insufficient technical		<b>315 310 200</b>	<b>379 500</b>	
6	Insufficient legal basis	<b>64 830</b>		<b>1 444 570</b>	
7	Non-compliance with			<b>358 400</b>	
8	Insufficient fulfilment	<b>7 000 000</b>		<b>27 000</b>	
9	Insufficient fulfilment			<b>259 800</b>	
10					
11					
12	=SUMMA(Table2)	=SUMMA(Table2 B:B)	=SUMMA(Table2 C:C)	=SUMMA(Table2 D:D)	
13	<b>No of Art refs Σ 128</b>	<b>12</b>	<b>3</b>	<b>113</b>	
14					
15	<b>Quoted Article</b>	<b>Sweden</b>	<b>UK</b>	<b>Spain</b>	
16	Art. 6		3	34	
17	Art. 5		2	25	
18	Art. 32			3	10
19	Art. 5 (1) f)				
20	Art. 13				
21	Art. 5 (1) c)		1		

När Table1 uppdateras med mer data nedan så får den både ytterligare en rad och ytterligare en kolumn, Germany.

Tabellen sträcker sig nu till och med rad 10 och formeln SUMMA(B5:B9) täcker inte längre hela kolumnen och behöver utökas till att omfatta rad tio.

Ett alternativ vore att istället summera hela B-kolumnen, SUMMA(B:B). Då kan tabellen förändra storlek utan att formeln missar något. Det spricker dock på att 1) vi får med data ur den gröna tabellen nedanför och 2) det skulle bli en cirkelberäkning eftersom cellen med formeln, B2, ingår i summaområdet.

Genom att summera skärningen mellan den blåa tabellen och kolumn B får vi med allt i den blå tabellen utan att summera något ur den gröna.

Formeln kan användas i B2 till och med E2 och kommer hela tiden att visa summan av den tabellkolumn som är rakt nedanför.

Summan för Spain är samma både innan och efter utökningen, och resultatet står strax ovanför rubriken. D2 respektive E2.

	A	B	C	D	E
1	=SUMMA(Table1)		=SUMMA(Table1 C:C)	=SUMMA(Table1 D:D)	=SUMMA(Table1 E:E)
2	<b>Fine € Σ 349 936 725</b>	=SUMMA(Table1 B:B)	<b>315 310 200</b>	<b>25 092 425</b>	<b>2 469 270</b>
3					
4	Type	Sweden	UK	Germany	Spain
5	Insufficient technical		<b>315 310 200</b>	<b>9 935 000</b>	<b>379 500</b>
6	Insufficient legal basis	<b>64 830</b>		<b>55 018</b>	<b>1 444 570</b>
7	Non-compliance with			<b>14 796 000</b>	<b>358 400</b>
8	Insufficient fulfilment	<b>7 000 000</b>		<b>245 407</b>	<b>27 000</b>
9	Insufficient fulfilment				<b>259 800</b>
10	Lack of appointment			<b>61 000</b>	
11					
12					
13	=SUMMA(Table2)	=SUMMA(Table2 B:B)	=SUMMA(Table2 C:C)	=SUMMA(Table2 D:D)	=SUMMA(Table2 E:E)
14	<b>No of Art refs Σ 164</b>	<b>12</b>	<b>3</b>	<b>36</b>	<b>113</b>
15					
16	Quoted Article	Sweden	UK	Germany	Spain
17	Art. 6		3		6
18	Art. 5		2		7
19	Art. 32			3	
20	Art. 5 (1) f)				
21	Art. 13				
22	Art. 5 (1) c)		1		

∴

SUMMA(Table1[Sweden]) måste autofyllas om den ska byta kolumn och inte summera Sweden i alla fyra. Inte som avsett.

SUMMA(Table1 B:B) kan kopieras till de andra tre på alla vanliga sätt.

- Autofyll B2 till E2
- Kopiera B2 och klistra in i C2:E2
- Markera B2:E2 och Fyll åt höger / Ctrl+R

## OMFEL(VÄRDE; VÄRDE\_OM\_FEL) / ISERROR

”Returnerar värde\_om\_fel om uttrycket är ett fel, i annat fall värdet för uttrycket.”

Skulle tabellen minska i sidled så att Table1 och den angivna kolumnen inte längre har någon skärningspunkt så blir det beräkningsfelet #SKÄRNING! / #NULL!.

Genom att lägga funktionen OMFEL runt summeringen kan modellen fås att visa ingenting om det händer. **OMFEL(SUMMA(Table1 D:D);"")** innebär att det inte visas någonting, "", om beräkningen skulle resultera i ett fel. Om den emellertid utan fel räknar fram ett resultat så visas det.

Nedan är OMFEL inlagd i D2 och E2.

	A	B	C	D	E
1	=SUMMA(Table1)	=SUMMA(Table1 B:B)	=SUMMA(Table1 C:C)	=OMFEL(SUMMA(Table1 D:D);"")	
2	<b>Fine €</b> <b>Σ 2 469 270</b>	<b>2 469 270</b>	<b>#SKÄRNING!</b>		
3					
4	<b>Type</b> <b>Spain</b>				
5	Insufficient technical	<b>379 500</b>			
6	Insufficient legal basis	<b>1 444 570</b>			
7	Non-compliance with	<b>358 400</b>			
8	Insufficient fulfilment	<b>27 000</b>			
9	Insufficient fulfilment	<b>259 800</b>			
10	Lack of appointment				

## SKILLNADEN MELLAN TVÅ LISTOR

Excels möjligheter att räkna med listor förenklar att jämföra två listor med varandra; Vad har försvunnit från första listan till andra, och vad har tillkommit?

Kolumn A innehåller alla länder vid en viss tidpunkt och vi vill hitta de som i A som saknas i kolumn D.

### VAD HAR FÖRSVUNNIT?

Till att börja med behöver vi ta fram de unika länderna.

#### UNIK(MATRIS) / UNIQUE

”Returnerar unika värden från ett intervall eller en matris”

Formeln UNIK(A:A) listar varje enskilt land i G2 och spiller nedåt. Inkluderar en nolla eftersom kolumnen innehåller tomma celler.

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount						
2	Portugal	400 000		Bulgaria	5 120		Country				
3	Austria	300		Norway	73 600		Portugal				
4	Czech Rep	388		Sweden	7 000 000		Austria				
5	Germany	20 000		Sweden	18 700		Czech Republic				
6	Bulgaria	500		Finland	128 500		Germany				
7	Austria	4 800					Bulgaria				
8							0				
9											

#### XMATCHNING(LETAUPPVÄRDE; LETAUPPVEKTOR) / XMATCH

”Returnerar den relativa positionen för ett objekt i en matris”

Genom att matcha de unika värden som listas i G2#, mot de som finns i kolumn D får vi en siffra som anger var i kolumn D värdet finns. Om värdet saknas blir svaret #SAKNAS!.

Formeln i H2 använder värdena i G2# som letauppvärde, och letar i vektorn/kolumnen D.

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount						
2	Portugal	400000		Bulgaria	5120		Country	1			
3	Austria	300		Norway	73600		Portugal	#SAKNAS!			
4	Czech Rep	388		Sweden	7000000		Austria	#SAKNAS!			
5	Germany	20000		Sweden	11200		Czech Rep	#SAKNAS!			
6	Bulgaria	500		Finland	128500		Germany	#SAKNAS!			
7	Austria	4800					Bulgaria	2			
8							0	#SAKNAS!			
9											

#### ÄRFEL(VÄRDE) / ISERROR

”Kontrollerar om ett värde är fel och returnerar SANT eller FALSKT”

De värden som saknas är de som finns i första listan men inte i den andra. Funktionen ÄRFEL(H2#) särskiljer de som finns från de som saknas så att resultatet blir endast sant eller falskt beroende på om värdet i H2# är fel.

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount						
2	Portugal	400 000		Bulgaria	5 120		Country	1	FALSKT		
3	Austria	300		Norway	73 600		Portugal	#SAKNAS!	SANT		
4	Czech Rep	388		Sweden	7 000 000		Austria	#SAKNAS!	SANT		
5	Germany	20 000		Sweden	18 700		Czech Rep	#SAKNAS!	SANT		
6	Bulgaria	500		Finland	128 500		Germany	#SAKNAS!	SANT		
7	Austria	4 800					Bulgaria	2	FALSKT		
8							0	#SAKNAS!	SANT		
9											

## FILTER(MATRIS; INKLUDERA) / FILTER

”Filtrera ett intervall eller en matris”

De värden som ska filtreras fram, matrisen, finns i G2#.

De som ska inkluderas är de rader som har SANT i i2#.

FILTER(G2#;i2#=SANT()) innebär att celler i G2# visas om motsvarande värde i i2# = SANT().

Sant = Sant är alltid sant och det räcker därför med bara i2# istället för i2#=SANT().

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount						
2	Portugal	400 000		Bulgaria	5 120		Country	1	FALSKT	Portugal	
3	Austria	300		Norway	73 600		Portugal	#SAKNAS!	SANT	Austria	
4	Czech Rep	388		Sweden	7 000 000		Austria	#SAKNAS!	SANT	Czech Republic	
5	Germany	20 000		Sweden	18 700		Czech Rep	#SAKNAS!	SANT	Germany	
6	Bulgaria	500		Finland	128 500		Germany	#SAKNAS!	SANT	0	
7	Austria	4 800					Bulgaria	2	FALSKT		
8							0	#SAKNAS!	SANT		
9											

## KOMBINERA BERÄKNINGARNA TILL EN CELL

Formlerna är nu enligt följande och kan, om man vill, kombineras till en enda.

```
J2=FILTER(G2#;i2#)
i2=ÄRFEL(H2#)
H2=XMATCHNING(G2#;D:D)
G2=UNIK(A:A)
```

G2=UNIK(A:A) och spiller ut för att visa alla unika värden. Formler som innehåller G2#, dvs G2 samt alla dess spillda värden, kan där ersättas med UNIK(A:A);

```
J2=FILTER(UNIK(A:A);i2#)
i2=ÄRFEL(H2#)
H2=XMATCHNING(UNIK(A:A);D:D)
```



Samma ersättning av H2 som ju är lika med XMATCHNING(UNIK(A:A);D:D) kortar ytterligare ett steg;

```
J2=FILTER(UNIK(A:A);i2#)
i2=ÄRFEL(XMATCHNING(UNIK(A:A);D:D))
```

Återstår endast att byta ut i2# mot ÄRFEL(XMATCHNING(UNIK(A:A);D:D)) så är allt gjort i en enda cell som kan flyttas till G2 och ges en rubrik.

```
FILTER(UNIK(A:A);ÄRFEL(XMATCHNING(UNIK(A:A);D:D)))
```

G2											
=FILTER(UNIK(A:A);ÄRFEL(XMATCHNING(UNIK(A:A);D:D)))											
	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount		Minus				
2	Portugal	400 000		Bulgaria	5 120		Portugal				
3	Austria	300		Norway	73 600		Austria				
4	Czech Rep	388		Sweden	7 000 000		Czech Republic				
5	Germany	20 000		Sweden	18 700		Germany				
6	Bulgaria	500		Finland	128 500						
7	Austria	4 800									
8											
9											

SUMMA.OMF(SUMMAOMRÅDE;VILLKORSOMRÅDE1;VILLKOR1) / SUMIFS

“Adderar de celler som anges av en given uppsättning villkor.”

Värdet som minskar är de saknade värdenas summa av kolumn B. Summaområdet är kolumn B, villkorsområdet är kolumn A där ländernas namn står, och villkoret är att de matchar de värden som saknas, dvs G2#.

```
SUMMA.OMF(B:B;A:A;G2#)
```

Excel utgår ifrån att jämförelsen ska vara ”lika med” G2#. Formeln kan förtydligas med att sammanfoga ett lika med-tecken<sup>8</sup> framför G2#;

```
SUMMA.OMF(B:B;A:A;"=" & G2#)
```

H2											
=SUMMA.OMF(B:B;A:A;G2#)											
	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount		Minus	425 488			
2	Portugal	400 000		Bulgaria	5 120		Portugal	400 000			
3	Austria	300		Norway	73 600		Austria	5 100			
4	Czech Rep	388		Sweden	7 000 000		Czech Rep	388			
5	Germany	20 000		Sweden	18 700		Germany	20 000			
6	Bulgaria	500		Finland	128 500						
7	Austria	4 800									
8											
9											

H1=SUMMA(H2#) och ger tydlig läsbarhet/spårbarhet eftersom summan per land visas i modellen.

<sup>8</sup> Även mindre än, större än eller ej lika med; "<", ">" respektive "<>" kan sammanfogas som en textsträng till en annan.

Vore beräkningen uppsatt endast för att summera minskningen så har vi ovan att H2=SUMMA.OMF(B:B;A:A;G2#) och det går följaktligen att förkorta till att

H1=SUMMA(SUMMA.OMF(B:B;A:A;G2#))

Och... eftersom G2= FILTER(UNIK(A:A);ÄRFEL(XMATCHNING(UNIK(A:A);D:D))) kan även dessa kombineras till en enda formel som visar summan av de länder som finns i kolumn A men saknas i kolumn D;

SUMMA(SUMMA.OMF(B:B;A:A;FILTER(UNIK(A:A);ÄRFEL(XMATCHNING(UNIK(A:A);D:D))))))

### VAD HAR TILLKOMMIT?

För att hitta länder som tillkommit kopieras G1:H2 till J1.

Rubriken i J1 ändras till Plus och formeln i J2 behöver korrigeras.

UNIK(A:A) som använts för att hitta unika värden i kolumn A har inga dollartecken och kolumnreferensen har följt med relativt åt höger.

Den blå kolumnen, D, är den som används för unika värden och har hamnat helt rätt.

Matchningen görs emellertid i kolumn G vilket är fel – kolumn D ska jämföras med länderna i första tabellen och den röda märkningen kan dras till att istället markera kolumn A.

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount		Minus	425 488		Plus	7 225 920
2	Portugal	400 000		Bulgaria	5 120		Portugal	400 000		D);G:G))	0
3	Austria	300		Norway	73 600		Austria	5 100		Bulgaria	5 120
4	Czech Rep	388		Sweden	7 000 000		Czech Rep	388		Norway	73 600
5	Germany	20 000		Sweden	18 700		Germany	20 000		Sweden	7 018 700
6	Bulgaria	500		Finland	128 500					Finland	128 500
7	Austria	4 800									

FILTER(UNIK(D:D);ÄRFEL(XMATCHNING(UNIK(D:D);A:A)))

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount		Minus	425 488		Plus	7 220 800
2	Portugal	400 000		Bulgaria	5 120		Portugal	400 000		D);A:A))	73 600
3	Austria	300		Norway	73 600		Austria	5 100		Sweden	7 018 700
4	Czech Rep	388		Sweden	7 000 000		Czech Rep	388		Finland	128 500
5	Germany	20 000		Sweden	18 700		Germany	20 000			
6	Bulgaria	500		Finland	128 500						
7	Austria	4 800									

### SORTERA(MATRIS) / SORT

”Sorterar ett intervall eller en matris.”

Om formeln därtill omsluts av funktionen SORTERA så kommer de filtrerade värdena i sorterad ordning.

SORTERA(FILTER(UNIK(D:D);ÄRFEL(XMATCHNING(UNIK(D:D);A:A))))

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount		Minus	425 488		Plus	7 220 800
2	Portugal	400 000		Bulgaria	5 120		Austria	5 100		Finland	128 500
3	Austria	300		Norway	73 600		Czech Rep	388		Norway	73 600
4	Czech Rep	388		Sweden	7 000 000		Germany	20 000		Sweden	7 018 700
5	Germany	20 000		Sweden	18 700		Portugal	400 000			
6	Bulgaria	500		Finland	128 500						
7	Austria	4 800									
8											
9											

## SKILLNADEN FÖR ALLA VÄRDEN

För att hitta de som saknas nyttjar formeln att ÄRFEL är sant.

`FILTER(UNIK(A:A);ÄRFEL(XMATCHNING(UNIK(A:A);D:D)))`

## ICKE(LOGISK) / NOT

“Ändrar sant till falskt eller falskt till sant”

För att hitta de som *inte* saknas utökas formeln med ICKE och parenteserna satta omkring ÄRFEL-delen.

`FILTER(UNIK(A:A);ICKE(ÄRFEL(XMATCHNING(UNIK(A:A);D:D))))`

	A	B	C	D	E	F	G	H	I	J	K
1	Country	Amount		Country	Amount			Delta	4 620		
2	Portugal	400 000		Bulgaria	5 120			Country			
3	Austria	300		Norway	73 600			Bulgaria	+4 620		
4	Czech Rep	388		Sweden	7 000 000			0			
5	Germany	20 000		Sweden	18 700						
6	Bulgaria	500		Finland	128 500						
7	Austria	4 800									
8											

	A	B	C	D	E	F	M	N	O	P	Q
1	Country	Amount		Country	Amount		Delta	4 620			
2	Portugal	400000		Bulgaria	5120		Country				
3	Austria	300		Norway	73600		Bulgaria	+4 620			
4	Czech Rep	388		Sweden	7000000		0				
5	Germany	20000		Sweden	11200						
6	Bulgaria	500		Finland	128500						
7	Austria	4800									
8											

Summeringsformeln i N2 motsvarar ”plusformeln” i K2 minus den tidigare ”minusformeln” i H2;

`K2=SUMMA.OMF(E:E;D:D;J2#)` där J2 är cellen till vänster om, ie vänster om K2.

`H2=SUMMA.OMF(B:B;A:A;G2#)` där G2 är cellen till vänster om.

K2-H2 blir ökningsbeloppet minus minskningsbeloppet och ger skillnaden.

`SUMMA.OMF(E:E;D:D;M2#)-SUMMA.OMF(B:B;A:A;M2#)`

	A	B	C	D	E	F	M	N	O	P	Q
1	Country	Amount		Country	Amount		Delta	4 620			
2	Portugal	400000		Bulgaria	5120		Country				
3	Austria	300		Norway	73600		Bulgaria	+4 620			
4	Czech Rep	388		Sweden	7000000		0				
5	Germany	20000		Sweden	11200						
6	Bulgaria	500		Finland	128500						
7	Austria	4800									
8											

Talformatet är ändrat till ett som visar plustecken och döljer nollvärden.

`"+"##0;[Röd]-##0;"`

För att dölja de tomma raderna i kolumnerna M och N – noga räknat, M2#:N2#, alltså från och med den dynamiska matrisen i M2 till och med den dynamiska i N2 – beräknas ett filter på området som endast tar med rader där det numeriska värdet i N2# inte är noll.

`FILTER(M2#:N2#; N2#<>0)`

Kompletteras med sortering som gör det enligt områdets andra kolumn och i fallande ordning.

`SORTERA(FILTER(M2#:N2#;N2#<>0);2;-1)`

	A	B	C	D	E	F	M	N	O	P	Q
1	Country	Amount		Country	Amount		Delta	4 620		Delta	4 620
2	Portugal	400000		Bulgaria	5120		Country			Bulgaria	+4 620
3	Austria	300		Norway	73600		Bulgaria	+4 620			
4	Czech Rep	388		Sweden	7000000		0				
5	Germany	20000		Sweden	11200						
6	Bulgaria	500		Finland	128500						
7	Austria	4800									
8											

Med mer data blir det tydligare att formeln spillt till två kolumner och att den sköter sorteringen enligt andra kolumnen, Q, från största till minsta belopp.

	A	B	C	D	E	F	M	N	O	P	Q
1	Country	Amount		Country	Amount		Delta	17 330 469		Delta	17 330 469
2	Czech Rep	388		Bulgaria	5110		Country			Italy	+16 352 000
3	Bulgaria	500		Spain	204000		Bulgaria	-3 187 730		Sweden	+6 976 270
4	France	5E+07		Spain	3000		Hungary	-199 311		Spain	+62 670
5	Czech rep	2330		Cyprus	10000		Poland	-925330		Belgium	+12 000
6	Germany	2500		Greece	15000		Italy	16352000		Norway	-9 600
7	Portugal	20000		Spain	3600		Norway	-9600		Cyprus	-101 000
8	Hungary	1560		Italy	27810000		Belgium	12000		The nethe	-160 000
9	Malta	5000		Italy	60000		Denmark	-333150		Hungary	-199 311
10	Hungary	1560		Spain	352470		Spain	62670		Denmark	-333 150
11	Bulgaria	500		Spain	1500		The nethe	-160000		Romania	-464 350
12	Bulgaria	27100		Italy	20000		Romania	-464350		Greece	-692 000
13	Czech rep	776		Romania	3000		Greece	-692000		Poland	-925 330
14	Czech rep	582		Italy	4000		Sweden	6976270		Bulgaria	-3 187 730
15	Hungary	3200		Spain	157500		Cyprus	-101000			
16	Hungary	3200		Spain	1500						
17	Portugal	2000		Bulgaria	5120						
18	Czech rep	10000		Greece	5000						
19	Portugal	2000		Spain	54000						

Summeringen i Q1 kan inte summera hela kolumn Q eftersom det skulle inkludera summan själv och generera en cirkelreferens.

Att begränsa till Q2:Q14 fungerar för stunden men blir inte säkert rätt när P2# ändrar storlek, alltefter förutsättningarna i indata.

Summera P2# skulle i det här fallet fungera pga att värdena i P-kolumnen endast är alfabetiska och ej påverkar summan numeriskt. Tabellen skulle emellertid kunna omfatta flera kolumner och det är därför bättre att begränsa summeringen till skärningen/överlappningen av P2# och kolumn Q. Mellanslag i områdesreferensen adresserar skärningen, dvs celler som ingår i båda områdena.

`SUMMA(P2# Q:Q)`

De tre tabellerna är här kombinerade till en enda som visar tillkomna värden, försvunna och skillnaden. Formeln har \$Y för att enkelt kopieras åt höger utan att referensen till den filtrerade och sorterade tabellen ändras relativt.

SUMMA(\$Y3# Z:Z)

	A	B	C	D	E	F	Y	Z	AA	AB	AC
1	GDPR	2019		GDPR	2020			plus	minus	delta	sum
2	Country	Amount		Country	Amount			233 100	-409 564 277	17 330 469	-392 000 708
3	Czech Republic	388		Bulgaria	5 110		Italy			+16 352 000	+16 352 000
4	Bulgaria	500		Spain	204 000		Sweden			+6 976 270	+6 976 270
5	France	50 000 000		Spain	3 000		Finland	+128 500			+128 500
6	Czech republic	2 330		Cyprus	10 000		Ireland	+75 000			+75 000
7	Germany	2 500		Greece	15 000		Spain			+62 670	+62 670
8	Portugal	20 000		Spain	3 600		Iceland	+29 600			+29 600
9	Hungary	1 560		Italy	27 810 000		Belgium			+12 000	+12 000
10	Malta	5 000		Italy	60 000		Malta		-5 000		-5 000
11	Hungary	1 560		Spain	352 470		Latvia		-7 000		-7 000
12	Bulgaria	500		Spain	1 500		Norway			-9 600	-9 600
13	Bulgaria	27 100		Italy	20 000		Czech Republi		-14 270		-14 270
14	Czech republic	776		Romania	3 000		Portugal		-24 000		-24 000
15	Czech republic	582		Italy	4 000		Lithuania		-61 500		-61 500
16	Hungary	3 200		Spain	157 500		Cyprus			-101 000	-101 000
17	Hungary	3 200		Spain	1 500		The Netherla			-160 000	-160 000
18	Portugal	2 000		Bulgaria	5 120		Hungary			-199 311	-199 311
19	Czech republic	10 000		Greece	5 000		Denmark			-333 150	-333 150
20	Portugal	2 000		Spain	54 000		Romania			-464 350	-464 350
21	Bulgaria	5 100		Norway	73 600		Greece			-692 000	-692 000
22	Poland	220 000		Spain	120 000		Poland			-925 330	-925 330
23	Hungary	36 275		Norway	36 800		Bulgaria			-3 187 730	-3 187 730
24	Bulgaria	510		Spain	51 600		Austria		-18 000 000		-18 000 000
25	Germany	80 000		Spain	107 800		Germany		-25 042 307		-25 042 307
26	Hungary	9 400		The nethes	525 000		France		-51 100 000		-51 100 000
27	Italy	50 000		Hungary	286		United Kingd		-315 310 200		-315 310 200
28	Poland	12 950		Poland	4 600						
29	Norway	120 000		Spain	60 000						
30	Czech republic	194		Italy	8 000						
31	Germany	1 400		Spain	7 200						
32	Lithuania	61 500		Poland	4 400						
33	Hungary	286		Spain	15 000						
34	Hungary	92 146		Denmark	21 000						
	Poland	2 000		Iceland	29 600						
	Sweden	400 000		Sweden							

### UNIKA VÄRDEN UR TVÅ LISTOR

Ett sätt att få ut unika värden ur båda listorna samtidigt vore att lägga den ena listan under den andra och sedan använda funktionen UNIK för hela kolumnen. Eg flytta lista 2 från D:E till A:100 och sedan använda UNIK(A:A).

En nackdel är att den första listan begränsas så att den max kan vara 99 rader. Lista två kan då flyttas till A500000 så har de en halv miljon rader vardera, men kalkylbladet blir då besvärligt att jobba med.

En kompakt väg att nå fram till en lista som består av två listors unika värden är att göra om dem till en xml-sträng och sedan bryta isär den.

XML-koderna är i sig valfria men här använder vi ett upplägg som motsvarar HTML. Länderna sammanfogas till en enda lång rad i en HTML-tabell. TEXTJOIN sammanfogar alla värden i A-kolumnen med </TD><TD> som tecken emellan.

"<TR><TD>"&TEXTJOIN("</TD><TD>";1;A:A;D:D)"&"</TD></TR>"

En mer komplett HTML-sträng skulle kunna inkludera de inledande och avslutande taggarna HTML, BODY och, framförallt, TABLE:

```
<HTML><BODY><TABLE><TR><TD>Country</TD><TD>Portugal</TD><TD>Austria</TD>
<TD>Czech
Republic</TD><TD>Germany</TD><TD>Bulgaria</TD><TD>Austria</TD><TD>Country
</TD><TD>Bulgaria</TD><TD>Norway</TD><TD>Sweden</TD><TD>Sweden</TD><T
D>Portugal</TD></TR></TABLE></BODY></HTML>
```

#### FILTRERAXML(XML; XPATH) / FILTERXML

Se [https://www.w3schools.com/xml/xpath\\_intro.asp](https://www.w3schools.com/xml/xpath_intro.asp) för introduktion till XML och XPath.

I det här sammanhanget räcker det emellertid med TR och TD för att kunna dela upp dem på den lägsta nivån, TD-noderna.

	A	B	C	D	E	F	G	H	I
1	Country	Amount		Country	Amount		Delta		
2	Portugal	400 000		Bulgaria	5 120		<TR><TD>Country</TD><TD>Portuga		
3	Austria	300		Norway	73 600				
4	Czech Republ	388		Sweden	7 000 000				
5	Germany	20 000		Sweden	18 700				
6	Bulgaria	500		Portugal	128 500				
7	Austria	4 800							
8									
9									
10									

Dela upp xml-strängen i enskilda celler;

```
FILTRERAXML("<TR><TD>"&TEXTJOIN("</TD><TD>";1;A:A;D:D)&"</TD></TR>";"/TD")
```

	A	B	C	D	E	F	G	H	I
1	Country	Amount		Country	Amount		Delta		
2	Portugal	400 000		Bulgaria	5 120		Country		
3	Austria	300		Norway	73 600		Portugal		
4	Czech Republ	388		Sweden	7 000 000		Austria		
5	Germany	20 000		Sweden	18 700		Czech Republic		
6	Bulgaria	500		Portugal	128 500		Germany		
7	Austria	4 800					Bulgaria		
8							Austria		
9							Country		
10							Bulgaria		

Begränsa till de unika

```
UNIK(FILTRERAXML("<TR><TD>"&TEXTJOIN("</TD><TD>";1;A:A;D:D)&"</TD></TR>";"/TD"))
```

	A	B	C	D	E	F	G	H	I
1	Country	Amount		Country	Amount		Delta		
2	Portugal	400 000		Bulgaria	5 120		Country		
3	Austria	300		Norway	73 600		Portugal		
4	Czech Republ	388		Sweden	7 000 000		Austria		
5	Germany	20 000		Sweden	18 700		Czech Republic		
6	Bulgaria	500		Portugal	128 500		Germany		
7	Austria	4 800					Bulgaria		
8							Norway		
9							Sweden		
10									

Komplettera med beräkning som visa skillnaden mellan listorna.

`SUMMA.OMF(E:E;D:D;G2#)-SUMMA.OMF(B:B;A:A;G2#)`

	A	B	C	D	E	F	G	H	I
1	Country	Amount		Country	Amount		Delta		
2	Portugal	400 000		Bulgaria	5 120		Country		
3	Austria	300		Norway	73 600		Portugal	-271 500	
4	Czech Republ	388		Sweden	7 000 000		Austria	-5 100	
5	Germany	20 000		Sweden	18 700		Czech Rep	-388	
6	Bulgaria	500		Portugal	128 500		Germany	-20 000	
7	Austria	4 800					Bulgaria	4 620	
8							Norway	73 600	
9							Sweden	7 018 700	
10									

Summera skillnaderna

`SUMMA(H2#)`

	A	B	C	D	E	F	G	H	I
1	Country	Amount		Country	Amount		Delta	+6 799 932	
2	Portugal	400 000		Bulgaria	5 120		Country		
3	Austria	300		Norway	73 600		Portugal	-271 500	
4	Czech Republ	388		Sweden	7 000 000		Austria	-5 100	
5	Germany	20 000		Sweden	18 700		Czech Rep	-388	
6	Bulgaria	500		Portugal	128 500		Germany	-20 000	
7	Austria	4 800					Bulgaria	+4 620	
8							Norway	+73 600	
9							Sweden	+7 018 700	
10									

## KOMMASEPARERADE FILER, CSV

Beroende på hur maskinmiljöerna är uppsatta så kan csv-filer komma med semikolon som separator (borde kallas scsv för semi colon separated values).



Med FILTERXML kan skiljetecknet, här kommatecken, bytas ut så att raden blir XML-kodad och kan separeras till enskilda celler.<sup>9</sup>

TRANSPONERA(FILTERXML("<TR><TD>" & BYT.UT(A1;" "; "</TD><TD>") & "</TD></TR>";"/TD"))

	A	B	C	D	E	F	G	H	I
1	Entity,alternative name,records lost,YEAR,story,SE-Entity		alternative records lo:	YEAR	story	SECTOR	METHOD	interesting	
2	AOL,American Online,92000000,2004,Jun 2004. A f	AOL	American	92000000	2004	Jun 2004.	web	inside job	#####
3	Ameritrade Inc.,online broker,200000,2005,Apr 20	Ameritrade Inc.	online bro	200000	2005	Apr 2005.	financial	lost device	#####
4	AOL,American Online,20000000,2006,Aug 2006. Du	AOL	American	20000000	2006	Aug 2006.	web	oops!	y
5	Swedish Transport Agency,,3000000,2017,Jul 2017	Swedish Transport Agency	#####	3000000	2017	Jul 2017.	li governmer	poor secur	y
6	Vårdguiden,Sweden's healthcare hotline,2700000,	Vårdguiden	Sweden's l	2700000	2019	Feb 2019.	healthcare	poor secur	y
7									

Skiljetecknet, gulmärkt, kan bytas ut mot semikolon om det är så filen ser ut.

## TRANSPONERA(MATRIS) / TRANSPOSE

Roterar en kolumn eller område från kolumnvis till radvis och vice versa.

Funktionen finns även i tidigare versioner av Excel men med dynamiska matriser har den blivit avsevärt mer både lättanvänd och behövd.

Används här då FILTERXML returnerar csv-raden som en kolumn/radvis.

<sup>9</sup> Denna formel är principiell och hanterar inte specialfall där kommatecken kan vara inneslutna av citationstecken och allt dem emellan skall tillhöra en enda cell.


DATUMFORMAT

Datum är ofta lite längre än Excels standardbredd på kolumnerna och istället för att visa en del av talet(!) visas endast hashtaggar, #.

	A	B	C
1	Datum	Country	Amount
2	#####	Portugal	400000
3	#####	Austria	300
4	#####	Czech repu	388
5	#####	Germany	20000

Text som är för lång trunkeras om cellen bredvid inte är tom.

Exempelvis cell B4 nedan;

Pekaren byter skepnad när den sätts över strecket mellan kolumnrubrikerna A och B. 

Det går att trycka ned och dra för att ändra bredd, eller dubbelklicka för att få den lagom anpassad av Excel.

Datum i Excel är alltså tal och inte text. Det är i princip antalet dagar sedan sekelskiftet 1900<sup>10</sup> och utifrån det har 1 jan 2000 heltalet 36526. Jämför 365 dagar per år x 100 år = 36500. Plus ett antal skottår.

INTERNATIONELLT DATUMFORMAT

Det finns en mängd olika sätt att skriva datum på och det är alltför ofta tvetydigt vilket datum som avses.

Exempelvis 03/08/12 är oklart när det inträffade. Står det istället 3 aug 2012 så blir det entydigt.

	A	B	C
1	Datum	Country	Sum Fine €
2	17 jul 2018	Portugal	400000
3	27 sep 2018	Austria	300
4	25 okt 2018	Czech Republic	388
5	21 nov 2018	Germany	20000

Excel har en stor mängd fördefinierade talformat och därtill kan du skapa helt egna.

Kom ihåg att ett talformat endast påverkar hur innehållet i en cell presenteras. Det påverkar inte innehållet.

Ett datum är ett heltal som motsvarar antalet dagar sedan en viss nollpunkt. Det kan följaktligen visas som datum, som heltal, som decimaltal etc alltefter behov.

<sup>10</sup> Macintosh Datumsystem utgår från 1 januari 1904 och om du får ca fyra år fel datum så är det förmodligen där felet ligger. Ställ om i *Arkiv: Excel-alternativ: Avancerat*.

Vid beräkning av den här arbetsboken:

- Uppdatera länkar till andra dokument
- Ange visad precision
- Använd Macintosh datumsystem
- Spara värden för fjärlänkar

Om ÅR(1) ger svaret 1904 så vet du att dokumentet är inställt på Macintosh datumsystem.


Windows datumsystem börjar fyra år tidigare än Macintosh' och är korrekt from 1 mars 1900. 29 feb 1900 återfinns endast i Excel - ej i verkligheten och ej i Outlook. Frånsett den avvikelser räknar Excel antalet dagar sedan årsskiftet 1900.

Unix datumsystem utgår från 1 jan 1970 och räknar sekunder istället för dagar.

Markera en cell och skriv ett datum.


Du kan exempelvis trycka Ctrl+Skift+Kommatecken så skriver Excel dagens datum.

*Startmenyn:* gruppen *Tal* har en knapp för att öppna dialogen för talformat. Ctrl+1 är ett annat sätt.



*Talformat: Kategori: Anpassat* listar en mängd färdiga talformat. Ovanför listan finns möjlighet att själv ändra det. D MMM ÅÅÅÅ visar ett datum som 3 aug 2012.

För tvåsiffrig dag i månaden, ange DD MMM ÅÅÅÅ så visas det istället som 03 aug 2012.



De fördefinierade talformaten är en bra grund att bygga vidare från.

3 aug 2012 formaterat som # ##0,00 visar 41 124,00. Vore talet 41 124,5 så vore det halvvägs till nästa dag, och följaktligen "klockan 12".

Decimaldelen anger tiden.

## INLEDANDE NOLLOR

Nummertecknet visar siffror om de behövs medan en nolla visar en siffra även om den inte behövs. Ett egendefinerat talformat med exempelvis åtta nollor, 00000000, skulle visa 00041124.

Genom att definiera nollorna i talformatet så ändras inte cellens värde. Den går fortfarande att räkna med som ett vanligt tal.

Skulle istället ett antal nollor läggas på som text, "000" & 41124, så gör Excel om resultatet till en textsträng – och den går inte att räkna med i exempelvis en summering. Är det ett tal så ändra utseendet med hjälp av talformat!

## SPECIELLA MÖJLIGHETER

Koden för ett talformat består av ett antal tecken som har fördefinierade betydelser. Resultatet kan också bli ett helt annat om man upprepar tecknet<sup>11</sup>.

"Koden" för talformat består av fyra olika delar;

plus; minus; noll; text

Talformatet **# ##0** avser endast plus-delen och används då även för minus och noll.

Negativa tal är i svensk Windows inställda på att ha minustecken längst fram och nollvärden kommer att visas med en enda nolla eftersom de övriga siffrorna är hashtaggar (nummertecken) som endast visas om de behövs.

**# ###** som ovan men visar ingenting om värdet är noll. Det är samma resultat som att ange

**# ##0;-# ##0**;

Den tredje parametern är vad som visas vid nollvärde och eftersom den är tom så visas ingenting.

Hakparenteserna visar att det går att specificera färg – som standardval finns att värden lägre än noll visas i rött. Funktionen kan användas till att färgmarkera om en cell innehåller text istället för numeriskt värde. Den fjärde parametern anger hur text ska visas och kan exempelvis byta färg.

**# ##0;-# ##0;0;[Blå]@**

<sup>11</sup>

0	Visar alltid en siffra och ger möjlighet att definiera inledande nollor. Antalet nollor är också det som visas vid nollvärde – om inget annat specificerats.
#	Visar en siffra endast om det behövs.
mellanslag	Standardinställd svensk tusentalsseparator. Om den finns med i talformatet blir stora tal lättare att läsa.
Å eller Y	Årtalssiffra. Y om engelsk/amerikansk installation.
M, MM, MMM, MMMM	Månadsnummer eller -namn.
D, DD	Dag i månaden. Utan eller med inledande nolla.
DDD, DDDD	Veckodag
t, tt, [t] eller h, hh, [h]	Timme. Den kan fås att räkna förbi 24 genom att ange [t]
m	Minuter. NB; litet m då stort M avser månad
s	Sekunder

Vill man förtydliga att det inte ska vara text i en cell så går det att undertrycka den faktiska texten och istället **visa** att det inte är tänkt så.

**# ##0;-# ##0;0;[Blå]"#N/A Text"**


Cellen kommer fortfarande att **innehålla** textvärdet men istället för textinnehållet **visas** #N/A Text.

Det är helt i linje med att en cell som **innehåller** =47110+1 kan **visas** som 47111, 47 111,00 eller **24 dec 2028** alltefter vilket format som används.

NB. Hakparenteserna kan användas till att ange vissa färger, men färgnamnen kan vara problem för Excel-modeller som används på flera språk.

Översättningsfunktionerna är inte alltid implementerade hela vägen och det är inte helt säkert att grön ändras till green.

Den nya funktionen XLETAUPP har fn inte fått sina parametrar översatta och TEXTJOIN har inte översatts till svenska.



## VISA VÄRDEN I TUSENTAL

Talformatet **# ##0** innehåller ett mellanslag/tusentalsseparator och resulterar i att stora tal delas vid tusen, miljon, miljard osv.

Sätts ett mellanslag sist så visas inga siffror under tusen. För att visualisera att talet trunckerats är det lämpligt att lägga till ett enkelt citationstecken eller k efter tusentalet.

Talformatet **# ##0 k** kommer att visa exempelvis 2000 som kort och gott 2k.

## SÄTTA EGNA GRÄNSER

Hakparenteserna kan användas till mer än färger. I grundläget avser "plus; minus; noll" att parameter ett styr värden över noll, parameter två värden under noll och parameter tre de som blir kvar, dvs noll.

Genom att ange egna gränser kan exempelvis färgerna styras till olika intervall. 5 och över i grönt, -3 och under i rött och resten i blått.

[Grön][>=5] 0;[Röd][<=-3] 0;[Blå] 0

## EXEMPEL

Första kolumnen visar vilket innehåll cellerna på raden har och varje kolumn har olika talformat som gör att värdet presenteras på olika sätt.

Koden för talformatet finns längst upp i varje kolumn. Observera dock att mellanslag som tusentalsseparator radbryter koden.

Cellen innehåller	<code>###0;[Röd] -###0;;@</code> Negative red. Zero hidden	<code>###0 '[Röd]-##0 ';0;@</code> Show as thousands	<code>[Grön][&gt;=5] 0;[Röd][&lt;=-3] -0;[Blå] 0</code> Custom limits	<code>+###0;[Röd]-##0; 0;[Röd] "#N/A Text"</code> Text not allowed	<code>;;;@</code> Text only
-4	-4	-0'	-4	-4	
-1	-1	-0'	- 1	-1	
0		0	0	0	
2,333333	2	0'	2,33	+2	
8,5	9	0'	9	+9	
10	10	0'	10	+10	
1200000	1 200 000	1 200'	1200000	+1 200 000	
Cookie	Cookie	Cookie	Cookie	#N/A Text	Cookie
2028-12-24	47 111	47'	47111	+47 111	

## MATRISBERÄKNINGAR

Control+Skift+Enter, i forum som LinkedIn Excel Black Belts ofta förkortat CSE, gör att en formel beräknas som med matriser. CSE har varit nödvändigt för att aktivera matrisberäkning i tidigare versioner (pre-dynamic array Excel).

{=SUMMA(B2:B5\*C2:C5)} har länge kunnat beräknas korrekt som matrisberäkning. Krullparenteserna förtydligar att formeln matats in med Ctrl och Skift nedtryckta.

Just denna möjlighet har bedömts så värdefull – och sällananvänd/okänd – att Microsoft skapade en ny funktion<sup>12</sup> just för detta, PRODUKTSUMMA(B2:B5;C2:C5). Den utför samma beräkning men utan att behöva matas in på annat sätt än som vanligt.

Senare versioner har nu fått matrisberäkningen som en automatisk tolkning och det fungerar utmärkt med bara SUMMA(B2:B5\*C2:C5).

Det går på samma sätt också att leta upp kombinationer där exempelvis B2 sammanfogas med B3 och sedan söks efter i kombinationen av kolumn A sammanfogad med kolumn C;

=XLETAUPP(E2&F2; data!B:B & data!C:C; data!A:A)

Det är i tidigare versioner av Excel nödvändigt att göras som manuellt angiven matrisberäkning;

{=INDEX(data!A:A; PASSA(E2&F2; data!B:B & data!C:C; 0))}

Matrisberäkningarna har emellertid fortsatt sin plats för mer avancerade formler som exempelvis kan loopa igenom antalet tecken och visa dem som ascii-koder<sup>13</sup>, göra om punktuppräknings och IP-nummer till numeriskt sorterbara etc. Framförallt i Excel-versioner tidigare än att de automatiskt hanterar dynamiska matriser (pre-dynamic array Excel).

<sup>12</sup> PRODUKTSUMMA / SUMPRODUCT

<sup>13</sup> {=TEXTJOIN("|";0;KOD(EXTEXT(\$A4;KOLUMN(FÖRSKJUTNING(\$A4;0;0;1;LÄNGD(\$A4)));1)))}

	A	B	C	D	E	F	G	H	I
1									
2									
3			=TEXTJOIN(" ";0;CODE(MID(\$A4;COLUMN(OFFSET(\$A4;0;0;1;LEN(\$A4)));1)))						
4	ASCII		65 83 67 73 73						

## INDEX(MATRIS;RAD;[KOLUMN])

”Returnerar ett värde eller en referens för cellen vid skärningen av en viss rad och kolumn i ett givet område”.

INDEX returnerar normalt en enskild cell utifrån värdena i rad och kolumn – tänk radnummer och kolumnnummer inom det angivna området/matrisen.

Om matrisen är endimensionell behövs endast rad varför kolumn står som ej obligatorisk parameter, inom hakparenteser.

I en tvådimensionell matris är både rad och kolumn obligatoriska. Om man inte anger vilken rad i kolumnen som önskas i retur, får man hela kolumnen. På samma sätt får man hela raden om man sätter noll eller utelämnar kolumnparametern.

Genom att ange noll som radnummer så refererar funktionen inte till en enskild rad i en enskild kolumn utan returnerar istället alla rader i kolumnen. Detsamma gäller omvänt att en nollad eller utebliven kolumn returnerar hela raden.

Som extremfall blir därmed resultatet lika för

=Table1 och

=INDEX(Table1;0;0)

## STATISKA MATRISER {5;6;4;7}{4\7\11}

Än så länge sällan använt men likväl en möjlighet som i om de dynamiska matriserna kan få ökad användning. Det är emellertid en ”hårdkodning” som inte påverkas av ändrade förutsättningar. Vanligen vill man ha dynamiska referenser som ändras när det skjuts in en kolumn eller en cell av annan anledning byter plats.

En formel inmatad med Control+Skift+Enter visar att den kalkyleras som en matrisberäkning genom att formeln omges av krullparenteser.

Manuellt angivna krullparenteser anger för Excel att mellan dem finns en matris med statiska värden. Om värdena separeras med **semikolon** så används de i **höjdled**. Om de separeras med **bakvänt snedstreck** \ så används de i **sidled**.

**NB**; i makron och med amerikanska inställningar används kommatecken för sidled.



	A	B	C	D	E	F
1						
2	Fine €	£ 324 844 300	7 064 830	315 310 200	2 469 270	
3						
4	Type	Sweden	UK	Spain		
5	Insufficient technical and organisational measures			315 310 200	379 500	
6	Insufficient legal basis for data processing	64 830			1 444 570	
7	Non-compliance with general data processing principles				358 400	
8	Insufficient fulfilment of data subjects rights	7 000 000			27 000	
9	Insufficient fulfilment of information obligations				259 800	
10	Lack of appointment of data protection officer					

Tabellen heter Table1\_txt

INDEX(B:D;4;1) hämtar alltid B4, i tabellen ovan Sweden. I området B:D, returnera fjärde radens första kolumn.

Sweden
--------

INDEX(B:D;4;0) hämtar alltid B4:D4 då nollan i kolumnargumentet gör att Excel returnerar *alla* kolumner på den angivna raden – vilket blir hela raden. Bestämmer man inte kolumnnummer så får man alla celler på områdesraden.

Sweden	UK	Germany
--------	----	---------

INDEX(B:D;4;{1\3}) hämtar kolumn ett och tre, dvs både Sweden och Spain. Resultatet returneras horisontellt.

Sweden	Germany
--------	---------

INDEX(B:D;4;{1;3}) hämtar kolumn ett och tre men returnerar vertikalt.

Sweden
Germany

Det går också att ange landsnamnen i xmatchning och få en matris med svaren;

XMATCHNING({"Germany"\"Sweden\"};Table1\_txt[#Rubriker])

4	2
---	---

Semikolon i en konstant matris avser vertikalt och bakvänt snedstreck<sup>14</sup> avser horisontellt.

Det går bra att använda båda teknikerna för att skapa ett tvådimensionellt område.

INDEX(B4:D9;{1;2;3};{3\2\1}) returnerar med kolumnerna i omvänd ordning.

Germany	UK	Sweden
9 935 000	315 310 200	
55 018	0	64 830

<sup>14</sup> Bakvänt snedstreck gäller svenska inställningar. Istället kommatecken i makron och med amerikanska inställningar

Kombinerat med xmatchning kan vi hårdkodat plocka ut namngivna kolumner ur en tabell.

`INDEX(Table1_txt[#Alla];{1;2;3};XMATCHNING({"Germany"\"Sweden\"};Table1_txt[#Rubriker]))`

Germany	Sweden
9 935 000	
55 018	64 830

Det går också att åstadkomma transponeringar genom att använda \ (sidled) som radparameter och ; (höjdled) för kolumn.

`INDEX(B4:D9;{3\2\1};{1;2;3})` returnerar transponerat genom att kolumnerna anges som radparameter.

64 830		Sweden
0	315 310 200	UK
55 018	9 935 000	Germany